/**
 * Copyright (c) 2009 ben rimbey
 *
 * Permission is hereby granted, free of charge, to any person obtaining a copy
 * of this software and associated documentation files (the 'Software'), to deal
 * in the Software without restriction, including without limitation the rights
 * to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the Software is
 * furnished to do so, subject to the following conditions:
 *
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 *
 * THE SOFTWARE IS PROVIDED 'AS IS', WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
 * THE SOFTWARE.
 *
 *
 * need help with this, or have an idea as to how we can make it better?
 * please email me: ben.rimbey@gmail.com
 * http://benrimbey.wordpress.com
 */

package org.puremvc.as3.multicore.utilities.asyncStartupUtils.patterns.command {

	import flash.utils.describeType;

	import org.puremvc.as3.multicore.interfaces.ICommand;
	import org.puremvc.as3.multicore.interfaces.INotification;
	import org.puremvc.as3.multicore.interfaces.INotifier;
	import org.puremvc.as3.multicore.patterns.observer.Notifier;
	import org.puremvc.as3.multicore.utilities.asyncStartupUtils.interfaces.IAsyncCommand;

	public class MacroAsyncCommand extends Notifier implements ICommand, INotifier {

		internal var commandList:Array = [];
		internal var cachedNotification:INotification;

		public function MacroAsyncCommand() {
			initializeMacroAsyncCommand();
		}

		public function execute(notification:INotification):void {
			cachedNotification = notification;
			spoolOffCommand();
		}

		protected function initializeMacroAsyncCommand():void {
			//must be overrode, for that is the puremvc way.
		}

		protected function addSubCommand(command:Class):void {
			checkClassAncestory(command);
			commandList.push(command);
		}

		protected function spoolOffCommand():void {
			var targetClass:Class = commandList.shift() as Class;
			var commandInstance:IAsyncCommand = new(targetClass)();
			commandInstance.initializeNotifier(multitonKey);
			commandInstance.completeCallBack = handleCommandComplete;

			commandInstance.execute(cachedNotification);
		}

		protected function handleCommandComplete():void {
			if (commandList.length > 0) {
				spoolOffCommand();
			}
		}

		protected function checkClassAncestory(classRef:Class):void {
			var def:XML = describeType(classRef);
			var name:String = def.@name;
			var found:Boolean = false;

			for each (var node:XML in def.factory.(@type == name).implementsInterface) {
				if (String(node.@type).indexOf("IAsyncCommand") > -1) {
					found = true;
					break;
				}
			}

			if (!found) {
				throw new Error("Class " + name + " must implement IAsyncCommand to be used with MacroAsyncCommand!");
			}
		}

	}
}